
PROJECT:

 MULTI-OBJECTIVE OPTIMIZATION AND ROBUST DESIGN


The objective of this project is to write a multi-objective genetic algorithm starting from a single objective version and to apply it, first on a simple test case and then for the robust design of the I-beam. 

Main references: 

[1]. 'Numerical Solution of Optimization Test-Cases by Genetic Algorithms' 

by N. Marco-Blaszka, JA Désidéri, INRIA report, 1999,  Available at:

 ftp://ftp.inria.fr/INRIA/publication/publi-pdf/RR/RR-3622.pdf

[2]. 'Robust design through the use of a Hybrid Genetic Algorithm' 

by K. Hacker, K. Lewis, proceedings of DETC’02 Available at:

http://does.eng.buffalo.edu/publications/publications/2002.Hacker.DETC.pdf

Problem description: 

1. Unconstrained Multi objective optimization :

The unconstrained multi-objective minimization problem can be formulated as follows (see [1], §3.2): 

Minimize fi(x) i=1..N

An important definition is the non dominance: x1 dominates x2 iff:

 

f1(x1)<f1(x2) and f2(x1)( f2(x2) or f2(x1)<f2(x2) and f1(x1)( f1(x2)

The non dominated individuals define the Pareto front.

In order to define a single cost function, a ranking by front is then done. Once the individuals have been ranked, they can be assigned the dummy fitness values:

f(x)=rank to which x belongs

A sharing principle is then added to maintain a good diversity in the Pareto front. First, for each element x a niche value is evaluated  (see formula in [1], page 9) and is then used to define the final cost function leading to the so-called ‘Non Dominated Sorted Genetic Algorithm’ (see [1], page 10).

2. A simple test case :

A simple test case is proposed in page 6 of [1]. It consists in maximizing two cost functions F1 and F2 with two parameters x and y in [-]. The optimal Pareto front for the unconstrained case  is depicted in Figure 6 of page 13.


3. Robust design of an I-beam :

The robust design of an I-Beam introduced in reference [2] has already been studied in the course and is now solved by using a multi-objective genetic algorithm. Here, instead of minimizing a barycentric combination of the mean and the variance near a given value of parameters, the problem consists in minimizing the same functions (mean and variance) but as two independent cost functions.

Work to do: 

1. Write with Scilab a multi-objective version of the genetic algorithm given in Annex 1. The main change will consist in defining a new cost function. The initial and final Pareto front will also have to be displayed. 

2. Test the multi objective version of the genetic algorithm on the test case given in paragraph 3.1 of [1] (unconstrained case, n=2, results on Figure 6a).

3. Apply the multi objective genetic algorithm to solve the robust I-Beam problem by minimizing two cost functions: the mean and the variance near a given value of parameters. Both functions are recalled in Annex 2. A display function of the I-Beam is given in Annex 3.

Deadline: May 1st

ANNEX 1 : example of a real valued genetic algorithm for Scilab
//

// selection: proportionate with ranking method 

// crossover: blend 

// mutation: non uniform

// 1-elitism

//

function y=rastrigin(x,n)  // the function to optimize

y=n+sum(x.^2-cos(2*%pi*x));

endfunction

// 

function evalpop=evaluation(X);  // evaluation of the whole population

[Npop,n]=size(X);n=n-1;

evalpop=zeros(Npop,1);

for i=1:Npop;

  evalpop(i)=rastrigin(X(i,1:n),n);

end

endfunction

//

function [Xpar,bestpar]=evalplus1elitism(Xpar,Xoldpar,Npop,gen) // evaluation plus 1-elitism

val=evaluation(Xpar);    

Xpar(:,n+1)=val;

[minval,index]=min(Xoldpar(:,n+1));  

bestpar=Xoldpar(index(1),:);

[minval,index]=min(val);

bestnewpop=Xpar(index(1),:);

if (bestnewpop($)>bestpar($))&(gen>1) then

    nrand=int(Npop*rand())+1;

    Xpar(nrand,:)=bestpar;

else 

    bestpar=bestnewpop;

end

endfunction

//

function Xsel=rankselect(X)    // ranking selection

Xsel=X;

[Npop,n]=size(X);n=n-1;

[Fsort,ind]=sort(X(:,n+1));

X=X(ind,:);

p=1:Npop;roulette=cumsum(p)/sum(p);

for i=1:Npop;

  u=rand();

  index=find(roulette<u);Nselect=max(index)+1;

  Xsel(i,:)=X(Nselect,:);

end

endfunction

//

function xoff=blcross(xpar)    // blend crossover

[np,n]=size(xpar);n=n-1;

xoff=xpar;

for i=1:n

   u=rand();

   xoff(1,:)=u*xpar(1,:)+(1-u)*xpar(2,:);

   xoff(2,:)=u*xpar(2,:)+(1-u)*xpar(1,:);

end

endfunction

//

function xoff=mutate(xpar,ngen,Ngen,xmin,xmax) // non uniform mutation

[np,n]=size(xpar);n=n-1;

b=5;

xoff=xpar;

for i=1:n

  u1=rand();u2=rand();

  if (u1<1/2) then

    xoff(1,i)=xpar(1,i)+(xmax(i)-xpar(i))*u2*(1-(ngen-1)/Ngen)^b;

   else

    xoff(1,i)=xpar(1,i)-(xpar(i)-xmin(i))*u2*(1-(ngen-1)/Ngen)^b;

   end;

end;

endfunction

//

//---------main program ---------------------

//
n=evstr(x_dialog('parameter number of the rastrigin function','2')); 

Npop=evstr(x_dialog('population number','30'));     

Ngen=evstr(x_dialog('generation number','50')); 

pc=evstr(x_dialog('crossover probability','0.9')); 

pm=evstr(x_dialog('mutation probability','0.6')); 

//

xmin=-5.12*ones(1,n);xmax=5.12*ones(1,n);

Xmin=ones(Npop,1)*xmin;Xmax=ones(Npop,1)*xmax;

u=rand(Npop,n);

pop=Xmin+(Xmax-Xmin).*u;  // random initialisation of the population

Xpar=[pop,zeros(Npop,1)];  

Xoff=Xpar;Xoldpar=Xpar;fmin=[];mineval=0;newval=zeros(Npop,1);Traj=[];

//

for gen=1:Ngen;

  Xpar=Xoff;

  [Xpar,bestpar]=evalplus1elitism(Xpar,Xoldpar,Npop,gen);  // evaluation plus 1-elitism

  Xoldpar=Xpar;

  fmin=[fmin,bestpar($)];

  //

  Xpar=rankselect(Xpar); // rank selection

  for i=1:2:(Npop-1)

    u1=int(Npop*rand())+1;u2=int(Npop*rand())+1;u3=rand();

    xpar=[Xpar(u1,:);Xpar(u2,:)];Xoff(i:(i+1),:)=xpar;

    if (u3<pc) then

     xoff=blcross(xpar);   //crossover

     Xoff(i:(i+1),:)=xoff;

    end

    for j=1:2

       u4=rand();

        if (u4<pm) then

          xoffl=mutate(xoff(j,:),gen,Ngen,xmin,xmax);  // mutation

          Xoff(i+j-1,:)=xoffl;

        end

    end

   end

end

[Xpar,bestpar]=evalplus1elitism(Xoff,Xoldpar,Npop,gen);  // evaluation plus 1-elitism

fmin=[fmin,bestpar($)];

//

//----------- results displays --------------

xset('window',0);

xbasc();

plot2d(0:Npop:(Npop*Ngen),fmin);

xtitle('convergence history','Neval','fmin');

//

disp('minimum obtained:');disp(bestpar(1:n));

disp('corresponding value by f:');disp(bestpar($));

ANNEX 2 : cost functions for the I-Beam

//

function y=beam(x)            // the cross section

n=prod(size(x));

y=2*x(2)*x(4)+x(3)*(x(1)-2*x(4));

endfunction

//

function y=penaltystress(x) // the penalty term on stress

n=prod(size(x));

y1=180000*x(1)/(x(3)*(x(1)-2*x(4))^3+2*x(2)*x(4)*(4*x(4)^2+3*x(1)*(x(1)-2*x(4))));

y2=y1+15000*x(2)/((x(1)-2*x(4))*x(3)^3+2*x(4)*x(2)^3)-16

y3=5000/(1/12*x(3)*(x(1)-2*x(4))^3+2/12*x(2)*x(4)^3+2*x(2)*x(4)*((x(1)-x(4))/2)^2)-0.1;

y=max(y2,0)^2+max(y3,0)^2;

endfunction

//

function y=penaltyboud(x)  // the penalty term on bounds

n=prod(size(x));

penalty=[max(10-x(1),0)^2,max(10-x(2),0)^2,max(9-10*x(3),0)^2,max(9-10*x(4),0)^2,max(2*x(4)-x(1),0)^2,min(80-x(1),0)^2,min(50-x(2),0)^2,min(5-x(3),0)^2,min(5-x(4),0)^2];

y=sum(penalty);

endfunction

//
K=1000;Nsamp=100;std=0.5;wu=1;ws=1-wu;

ech=[];

for ks=1:Nsamp

u=std*(2*rand(1,n,'u')-1);

   ech=[ech,beam(x+u)+K*(penaltyboud(x+u)+penaltystress(x+u))];

end

cost1=mean(ech);                 // the mean

cost2=st_deviation(ech)^2;       // the variance 

// 

ANNEX 3 : I-Beam display
//

function plotIbeam(x);

//

X=[0,x(2),x(2),x(2)/2+x(3)/2,x(2)/2+x(3)/2,x(2),x(2),0,0,x(2)/2-x(3)/2,x(2)/2-x(3)/2,0,0]

Y=[0,0,x(4),x(4),x(1)-x(4),x(1)-x(4),x(1),x(1),x(1)-x(4),x(1)-x(4),x(4),x(4),0];

//

plot2d(X,Y,1,rect=[0,0,50,80]); 

endfunction

//

